Como utilizar o programa Maxima – Parte I
Prof. Rui – DMA – UEM
Fevereiro de 2014

Instalação

O programa Maxima é um programa gratuito e possui versões para Windows e Linux.

Usuários Linux
Os usuários de Linux podem procurar versões para sua distribuição diretamente no site do Projeto Sourceforge (www.sourceforge.net) e ver as instruções.

Usuários Ubuntu
No caso particular dos usuários da distribuição UBUNTU, o software wxMaxima pode ser encontrado para instalação na aba de programas Central de Programas do Ubuntu.



Nessa tela, escolha Ciência e Engenharia e mantenha-se conectado à Internet.



Em seguida escolha Matemática.



Ao escolher wxMaxima, verá uma descrição do software e a opção para “instalar”. Faça essa escolha e informe sua senha de administrador. Após alguns minutos o Maxima e sua interface wxMaxima estarão instalados.



Usuários Windows

Os usuários do sistema Windows podem baixar a versão 12.01.0 de janeiro de 2012 num dos links a seguir:

http://andrejv.github.com/wxmaxima/

http://sourceforge.net/projects/maxima/files/

Após baixar o arquivo você verá em sua área de downloads o seguinte ícone.

Execute-o, dê clique-duplo sobre o ícone. Você verá um aviso de segurança.



Confie no editor, clique em “Executar”. Será mostrada a caixa de escolha de idiomas.



Escolha o idioma Português. Provavelmente ele já estará selecionado. Clique em OK. Você verá a tela de boas vindas.



Clique em Avançar. Você verá a tela com os termos da licença.



Escolha a opção "Eu aceito os termos do Contrato" e clique em Avançar. Você verá uma tela de informação.



Clique em Avançar novamente. Será sugerida uma pasta para a instalação.



Aceite a pasta sugerida, clique em Avançar. Será mostrada uma tela com os itens a serem instalados. Não altere nada, clique em Avançar.



O programa fará a sugestão de nome de pasta para o menu Iniciar.



Aceite a sugestão, clique em Avançar. O programa mostrará o resumo da instalação.



Aceite, clique em Instalar.

Aguarde a instalação. Ao final da instalação o programa mostrará informações.



Escolha Avançar.

Depois será mostrada a tela de finalização.



Clique em Concluir. Você verá o ícone do Maxima em seu Desktop.

Execute o programa.

Dependendo de seu sistema operacional e de sua conexão com a Internet, ao executar o programa pela primeira vez poderão aparecer as seguintes telas:





Clique em Permitir.

Primeiros Passos

A janela do Maxima é semelhante à seguinte figura. Os usuários do sistema Ubuntu encontrarão o menu da interface wxMaxima incorporado à área de trabalho.



O Maxima indica ao usuário que está pronto para a entrada de comandos.
As operações aritméticas são realizadas mediante a utilização dos símbolos:

+ para a adição;
- para a subtração;
* para a multiplicação;
/ para a divisão e
^ para a exponenciação.

Por exemplo, escreva 123 + 567 e tecle Shift + Enter.
Ao começar a digitar o usuário verá que os caracteres aparecem numa célula na tela do software.



Ao teclar simultaneamente Shift + Enter o Maxima fornecerá uma saída em sua janela.

```
(%i1) 123+456;
(%o1) 579

(%i2)
```

Isso mostra que o programa "entendeu" o comando e o realizou. A resposta do programa é 579. O símbolo (%i2) indica que o programa está disponível para o próximo comando.
É possível colocar o cursor na linha já compilada e alterá-la. Por exemplo, se houve um erro de digitação e o cálculo deveria ser 1230 + 456, deve-se fazer o seguinte: Dar um clique-duplo sobre a linha (%01), situar o cursor após o algarismo 3, digitar o algarismo 0 e apertar simultaneamente as teclas Shiftl + Enter.

A saída será corrigida e a tela mostrará o seguinte.

```
(%i2) 1230+456;
(%o2) 1686

(%i3)
```

O Maxima refez o cálculo e fornece outra saída indicando o resultado.

O Maxima aguarda que o usuário inicie a digitação de outro comando para mostrar a célula de compilação, mas é possível forçar o aparecimento de novas células com a escolha do caminho "Cell – Insert Input Cell".

Force o aparecimento de uma nova célula. Escreva, por exemplo, 44 + 55, em seguida digite o sinal de ponto e vírgula no final do comando e aperte as teclas Shift + Enter simultaneamente. Você verá a seguinte saída.

```
(%i3) 44+55;
(%o3) 99
```

Perceba que a interface wxMaxima é amigável e completa a digitação com o sinal “;” ao final da linha. Esse sinal é o sinal de final de comando e pode ser também escrito pelo usuário sempre que quiser. Por exemplo, se o usuário quiser calcular duas operações de uma só vez pode digitar os sinais “;” separando as tarefas. Veja como na ilustração a seguir.

```
(%i3) 22+44;77+11;
(%o3) 66
(%o4) 88
```

Ao encerrar os cálculos você pode salvar o arquivo de trabalho mediante o uso do botão no menu principal, ou escolhendo o caminho Arquivo-Salvar. Os arquivos de trabalho do Maxima possuem extensão “wxm”, e podem ser reabertos e reeditados posteriormente.

A definição de funções se faz mediante a sintaxe “f(x) := expressão”. Deve-se utilizar o sinal de dois pontos seguido do sinal de igualdade. Após a compilação do comando de definição a função fica na memória de trabalho e pode ser utilizada para outros cálculos com a sintaxe “f(ponto)”. Na figura a seguir ilustramos a definição da função “f” e depois o cálculo de “f” no ponto 3.14159.

```
(%i8) f1(x):=sin(x);
(%o8) f1(x):=sin(x)

(%i9) f1(3.14159);
(%o9) 2.65358979335272610^-6
```

O menu do Maxima traz muitas opções de comandos, um deles calcula somatórias.
Por exemplo, para se calcular a somatória de termos 1/n, com n iniciando em 1 e finalizando em 10 faz-se o seguinte:

Escolhe-se no menu o caminho “Cálculo-Calcular soma” e será mostrada a seguinte caixa de diálogo:



Deve-se preenchê-la da seguinte maneira:



Ao se clicar no botão OK, a somatória será calculada e o resultado mostrado na tela.
O leitor verificará que o valor da somatória está expresso na forma de fração.

(%i1) sum(1/n, n, 1, 10), simpsum;

(%o1) $\frac{7381}{2520}$

Caso se queira o valor expresso em casas decimais deve-se alterar o comando e escrever a expressão "1.0/n" no local "1/n". Para isso usa-se um duplo-clique sobre a expressão mostrada na tela e depois de alterá-la aperta-se Ctrl + Enter.

(%i2) sum(1.0/n, n, 1, 10), simpsum;

(%o2) 2.928968253968254

Uma utilização muito frequente do Maxima é a que permite a visualização de esboços de gráficos.
Vejamos como fazer isso.
Defina primeiramente uma função.
Numa célula nova, escreva, por exemplo, o seguinte: f(x) : = sin(x)/x . Aperte Ctrl + Enter.
Você verá a seguinte saída na tela:

(%i1) f(x):=sin(x)/x;

(%o1) $f(x):=\frac{\sin(x)}{x}$

Escolha no menu o caminho "Gráficos – Gráfico 2d". Você verá uma janela, preencha o campo Expressão(ões) com f(x) e aperte OK.



Você verá a seguinte saída gráfica.



A linha na cor azul mostra o comando necessário para traçar gráficos no Maxima, caso você queira traçar gráficos mediante digitação de comandos deverá escrever comando semelhante ao mostrado nessa linha. Em seguida o Maxima apresenta o esboço do gráfico da função sobre o intervalo [-5, 5].

Analisando o traço do gráfico pode-se perceber que quando x se aproxima de x = 0 o valor funcional se aproxima de y = 1. Será possível que f(0) = 1? Para testar essa possibilidade digita-se numa nova célula o seguinte: f(0) e aperta-se Ctrl + Enter.

A saída mostrada é:

```
(%i4) f(0);
expt: undefined: 0 to a negative exponent.
#0: f(x=0)
 -- an error. To debug this try: debugmode(true);
```

Isso indica que x = 0 não pertence ao domínio da função, logo não existe valor f(0).

O caminho "Gráficos – Gráfico 2d" pode ser usado mesmo sem se definir uma função previamente, para isso, basta escrever a definição no campo Expressão(ões). Caso já existam duas funções definidas na memória, o caminho "Gráficos – Gráfico 2d" pode ser usado para traçar dois (ou mais) gráficos no mesmo quadro. Para isso, basta escrever os nomes das funções (ou as definições) separadas por vírgulas como mostra a figura a seguir.



A saída será semelhante à seguinte ilustração.



Outra utilidade do Maxima é a resolução de equações. Essa funcionalidade pode se acessada pelo caminho "Equações – resolver". Escreve-se a equação desejada no campo apropriado e obtém-se a resposta. No seguinte exemplo, escrevemos a equação f(x) = g(x).



A resposta dada pelo Maxima é:

```
(%i7) solve([f(x)=g(x)], [x]);
`solve' is using arc-trig functions to get a solution.
Some solutions will be lost.
(%o7) [x=0, x=-1, x=1]
```

Percebemos que a solução x = 0 não pode ser considerada, já que a função $f(x) = \frac{\sin(x)}{x}$ não está definida nesse ponto. Podemos ver isso também olhando os dois gráficos, evidentemente eles não se interceptam para x = 0.

As outras soluções, x = -1 e x = 1, parecem estar de acordo com as intersecções dos gráficos, mas existem outras soluções. O Maxima avisa que podem existir outras soluções. Percebemos que existe uma intersecção dos gráficos sobre o intervalo [2, 4]. Ela pode ser encontrada mediante a resolução numérica. Para fazer isso, deve-se escolher o caminho "Equações – Encontrar raiz". O preenchimento dos campos deve seguir a seguinte figura.



A resposta dada pelo Maxima é:

```
(%i8)  find_root(f(x)=g(x), x, 2, 4);
(%o8)  3.141592653589793
```

Percebemos, então, que a solução aproximada da equação f(x) = g(x) entre x = 2 e x = 4 é x ~ 3.14159. Se prestarmos atenção à equação considerada, veremos que quando x = π, o valor de seno é zero e a equação reduz-se a 0 = 0. Essa equação considerada aqui neste exemplo possui, na verdade infinitas soluções. Isso pode ser avaliado mediante exploração dos gráficos alterando-se o intervalo do traçado para valores cada vez maiores de x (e de –x).

Com essas rápidas instruções o usuário já pode fazer uso do Maxima para seus estudos de matemática.

Obs: Algumas configurações podem ser feitas para melhorar a interface com o usuário.

Por exemplo, pode-se escolher que apertando apenas a tecla Enter os comandos sejam calculados, isso pode ser feito mediante escolha "Editar - Configurações", na janela deve-se selecionar o item "Enter calcula células".

Outra facilidade oferecida ao usuário é a colocação de painéis com atalhos no lado esquerdo da janela do software. Isso se faz mediante escolha "Maxima – Painéis". Basta selecionar os painéis que se deseja.

FIM

xxxxxxxxxxxxxxxxx
OBS: Com essa introdução já se pode pensar em explorar números reais, proximidades, somatórias, sequências, e grandezas infinitamente pequenas e infinitamente grandes.
xxxxxxxxxxxxxxxxx

Como utilizar o programa Maxima – Parte II

Prof. Rui – DMA – UEM

Fevereiro de 2014

Ajuste de casas decimais

Por padrão o Maxima trabalha com quinze (15) casas decimais, mas existe a possibilidade de alteração da quantidade de casas consideradas mediante o uso do comando "fpprec:___;". (abreviação de flutuant point precision). Outra maneira de ajustar a quantidade de casas decimais é mediante o caminho Numérico – Ajustar precisão... no menu. Na janela que aparecerá, o usuário poderá escolher a quantidade de casas, como por exemplo, 37 casas.



O Maxima inserirá o comando "fpprec:37" e fará a alteração.

```
(%i1) fpprec:37;
(%o1) 37
```

Após a compilação do comando pode-se usar o comando "bfloat(__)" como no exemplo a seguir.

```
(%i2) bfloat(2/13);
(%o2) 1.538461538461538461538461538461538462b-1
```

A notação da saída informa que o número mostrado está multiplicado por 10^{-1}.
Outra maneira de se calcular aproximações numéricas é mediante a entrada da expressão, por exemplo $\frac{2}{13}$, no campo de Entrada. Depois, deve-se escolher o caminho "Numérico – Para bigfloat". O Maxima fará a aplicação do comando "bigfloat(__)" na expressão da linha anteriormente exibida na tela.

O uso do comando "float(__)" ainda utiliza a quantidade padrão de casas decimais como pode ser verificado em seguida. Ele também pode ser usado mediante o caminho "Numérico – Para float" .

```
(%i3) float(2/13);
(%o3) 0.15384615384615
```

Alguns números especiais utilizados na matemática, como e, π, podem ser utilizados mediante digitação dos comandos “%e” e “%pi”. Para obter aproximações decimais desses números pode-se usar o comando “float(___)”.

```
(%i1) %e;
(%o1) %e
(%i2) float(%e);
(%o2) 2.718281828459045
(%i3) %pi;
(%o3) %pi
(%i4) float(%pi);
(%o4) 3.141592653589793
```

O número complexo $i = \sqrt{-1}$ também pode ser utilizado com a chamada “%i” conforme exemplos mostrados a seguir. Por exemplo, podemos entrar com a expressão $(2+i)^3$ no campo de entrada e obteremos a saída:

```
(%i1) (2+%i)^3;
(%o1) (%i+2)^3
```

O Maxima possui vários painéis com comandos. Escolha o caminho “Maxima – Painéis – Matemática Geral”. Você verá que um painel aparecerá no lado esquerdo da janela do software. Você pode até arrastá-lo para fora da janela principal.

Para expandir a expressão anterior, basta clicar no botão 

do painel anteriormente mostrado. O Maxima calculará a expansão da expressão anterior.

```
(%i2) expand(%);
(%o2) 11%i+2
```

O Maxima escreverá, com sua notação, o resultado $11i+2$. O mesmo resultado pode ser alcançado mediante digitação direta do comando:

```
(%i1) expand((2+%i)^3);
(%o1) 11 %i + 2
```

Algumas funções matemáticas já se encontram pré-programadas no Maxima, um resumo de tais comandos encontra-se no quadro a seguir.

abs(x)	$\lvert x \rvert$
sqrt(x)	$\sqrt{x}$
exp(x)	e^x
log(x)	$\ln(x)$
sin(x); cos(x); tan(x)	$\operatorname{sen}(x)$; $\cos(x)$; $\operatorname{tg}(x)$

Se houver a necessidade de se trabalhar com o logaritmo de base 10, pode-se definir uma nova função mediante a digitação de comando semelhante ao seguinte no campo de Entrada.

```
(%i1) logbase10(x):=log(x)/log(10.0);
```

(%o1) $\text{logbase10}(x) := \dfrac{\log(x)}{\log(10.0)}$

Com essa função armazenada na memória do Maxima, pode-se calcular valores do logaritmo base 10.

```
(%i2) logbase10(2.0);
(%o2) 0.30102999566398
```

Sequências Numéricas

A criação de sequências numéricas no Maxima se faz mediante o uso do caminho "Álgebra – Criar lista...". Na janela "Criar lista" deve-se escrever a expressão, o nome da variável e a variação numérica da variável. No exemplo seguinte escolheu-se criar uma lista (ou sequência) com termo geral dado por $\dfrac{2k}{200}$ com k percorrendo os naturais entre 1 e 10.



O Maxima retorna a seguinte saída:

```
(%i1) makelist(2*k/200, k, 1, 10);
```

(%o1) $[\frac{1}{100},\frac{1}{50},\frac{3}{100},\frac{1}{25},\frac{1}{20},\frac{3}{50},\frac{7}{100},\frac{2}{25},\frac{9}{100},\frac{1}{10}]$

Pode-se armazenar uma sequência (ou lista) com um nome depois de criá-la, para fazer isso, deve-se escolher qual será o nome, por exemplo “A”, usar o sinal de atribuição “:” e em seguida o sinal de porcentagem “%”, que indica ao Maxima que a atribuição deve ser feita na saída anterior.

```
(%i2) A:%;
```

(%o2) $[\frac{1}{100},\frac{1}{50},\frac{3}{100},\frac{1}{25},\frac{1}{20},\frac{3}{50},\frac{7}{100},\frac{2}{25},\frac{9}{100},\frac{1}{10}]$

Aplicado o comando anterior, a lista está armazenada com o nome “A”.
Pode-se criar outra lista de nome “B” e adicioná-las elemento a elemento, desde que tenham o mesmo comprimento.

```
(%i3) makelist(k/100, k, 1, 10);
```

(%o3) $[\frac{1}{100},\frac{1}{50},\frac{3}{100},\frac{1}{25},\frac{1}{20},\frac{3}{50},\frac{7}{100},\frac{2}{25},\frac{9}{100},\frac{1}{10}]$

```
(%i4) B:%;
```

(%o4) $[\frac{1}{100},\frac{1}{50},\frac{3}{100},\frac{1}{25},\frac{1}{20},\frac{3}{50},\frac{7}{100},\frac{2}{25},\frac{9}{100},\frac{1}{10}]$

Escrevendo-se A+B no campo de Entrada pode-se obter a nova sequência.

```
(%i5) A+B;
```

(%o5) $[\frac{1}{50},\frac{1}{25},\frac{3}{50},\frac{2}{25},\frac{1}{10},\frac{3}{25},\frac{7}{50},\frac{4}{25},\frac{9}{50},\frac{1}{5}]$

Pode-se retormar um elemento específico de uma lista, por exemplo, o elemento B_3:

```
(%i6) B[3];
```

(%o6) $\frac{3}{100}$

Se houver uma função definida na memória será possível calcular as imagens da função aplicada nos elementos da sequência. Por exemplo, podemos calcular a raiz quadrada dos elementos da sequência A.

```
(%i7) map(sqrt,A);
```

$$\text{(\%o7)}\quad \left[\frac{1}{10},\frac{1}{5\sqrt{2}},\frac{\sqrt{3}}{10},\frac{1}{5},\frac{1}{2\sqrt{5}},\frac{\sqrt{3}}{5\sqrt{2}},\frac{\sqrt{7}}{10},\frac{\sqrt{2}}{5},\frac{3}{10},\frac{1}{\sqrt{10}}\right]$$

O mesmo resultado seria obtido se utilizássemos o caminho “Álgebra – Mapear a lista...” e preenchêssemos a janela da seguinte maneira:



Outra operação que se pode realizar sobre sequências (listas) é a aplicação de comandos. Na figura a seguir mostra-se a janela que aparece quando se escolhe o caminho “Álgebra – Aplicar a lista...”



A aplicação do comando de adição de termos fornece um retorno como o seguinte:

```
(%i3) apply("+", A);
```

$$\text{(\%o3)}\quad \frac{11}{20}$$

A manipulação de sequências no Maxima sofre restrições quando a quantidade de termos é muito grande. O comando de criação de lista “makelist(1,k,1,3200)” deveria ser compilado e o Maxima deveria escrever uma sequência com todos os 3200 termos idênticos a 1, mas o retorno não é esse!

```
(%i1) makelist(1,k,1,3200);
 << Expressão longa demais para ser exibida! >>
```

O Maxima consegue criar a lista se o número de termos for 3100, mas com 3200 termos não será possível fazer isso. Após criar a lista de 3100 termos se houver interesse no cálculo da soma desses termos o Maxima consegue realizar tal operação.

```
(%i3) apply("+", %);
(%o3) 3100
```

Mas se houver a necessidade de investigar o comportamento de somas de termos de sequências com um número grande de elementos deve-se usar o caminho “Cálculo – calcular soma...”.



O retorno desse comando é evidentemente 5372, como mostrado a seguir.

```
(%i4) sum(1, k, 1, 5372), simpsum;
(%o4) 5372
```

O comando “sum” ilustrado anteriormente executa a somatória $\sum_{i=1}^{5372} 1$.

Para executar produtórias, como por exemplo $\prod_{i=1}^{7} \frac{i}{i+1}$, deve-se utilizar o caminho “Cálculo – calcular produto...”.



O resultado é exibido da seguinte maneira:

(%i5) `product(i/(i+1), i, 1, 7);`

(%o5) $\frac{1}{8}$

Uma opção para se trabalhar com sequências é considerar que uma sequência é uma particular função cujo domínio é o conjunto dos números naturais.

Por exemplo, para se estudar o comportamento da sequência de termo geral igual a $a_n = \dfrac{n-2}{n^2-n}$ para $n \geq 2$, deve-se definir a função como mostrado a seguir.

```
(%i1) a(n):=(n-2)/(n^2-n);
(%o1) a(n):=(n-2)/(n^2-n)
```

Assim podemos analisar os termos de índice 5000, 7000, etc.

```
(%i3) float(a(5000));
(%o3) 1.9995999199839967 10^-4
```

```
(%i4) float(a(7000));
(%o4) 1.4283673177800909 10^-4
```

Representação gráfica de sequências numéricas

Para representar graficamente uma sequência mediante colocação de pontos no R2 é necessário criar uma sequência de pontos de R2 no Maxima, isso pode ser feito da seguinte maneira: Escolha "Álgebra – Criar lista..." e escreva a formulação geral dos pontos de R2 da forma (n, a_n) de acordo com o exemplo da figura a seguir.



Observe que pontos de R2 são denotados com colchetes no Maxima. Ao compilar o comando anterior o Maxima mostrará a sequência de pontos. Coloque o cursor na linha de comando mostrada pelo Maxima e insira um nome para a sequência e o sinal ":".

```
--> pts: makelist([n,(n-2)/(n^2-n)], n, 2, 20);
```

Compile o comando novamente (Ctrl + Enter).
Depois, para exibir a sequência de pontos de R2 cujo nome é "pts" basta digitar o comando:

```
--> plot2d([discrete,pts],[style,points]);
```

O Maxima criará numa janela separada o gráfico semelhante à seguinte figura.



Para interpretar essa representação devemos observar que no eixo horizontal temos a indexação dos termos da sequência numérica e o valor do termo a_n deve ser observado no eixo vertical.

Para observar o comportamento dos termos de índice 20 até índice 200, digitamos o comando "pts:makelist([n,a(n)], n, 20, 200);" ou colocamos o cursor na linha do comando anterior e editamos o intervalo. Após a criação dos pontos de R2 compilamos o comando "plot2d([discrete,pts],[style,points]);". Veremos o seguinte:



Observamos que à medida que o índice "n" aumenta os valores a_n são positivos e decrescentes. Ao analisar o comportamento dos termos da sequência a_n com n entre 8000 e 8500 observamos que a sequência continua positiva e decrescente, já na ordem de décimos de milésimos.



Existe a possibilidade de mostrar gráficos de pontos dentro do ambiente do Maxima, isso deve ser feito de maneira análoga ao da seguinte figura:



É claro que a exploração do comportamento de uma sequência numérica via sua representação gráfica tem vantagens e desvantagens. Uma das desvantagens é a dificuldade de criação de listas com mais de 3000 pontos no Maxima.

Pequenas rotinas para criar sequências

O Maxima permite que utilizemos pequenas rotinas de programação para criar sequências numéricas. Elas podem ser criadas mediante alteração do seguinte exemplo:

```
(%i18) for n:2 thru 10 do display((n-2)/(n^2-n))$
```

Como utilizar o programa Maxima – Parte III
Prof. Rui – DMA – UEM
Fevereiro de 2014

Matrizes

Uma matriz pode ser inserida no Maxima mediante escolha do caminho “Álgebra – introduzir matriz...”.



Essa janela oferece a possibilidade de escolha da dimensão, do nome, e se se quer uma matriz geral, diagonal, simétrica ou antisimétrica. Após a escolha do tipo de matriz é apresentada uma janela ao usuário para a digitação dos elementos da matriz.



A saída gráfica do Maxima será:

```
(%i3) A: matrix(
       [1,2,3],
       [4,5,6],
       [a,b,c]
      );
```

$$(\%o3)\quad \begin{bmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \\ a & b & c \end{bmatrix}$$

Pode-se multiplicar uma matriz por um escalar mediante o uso do campo de Entrada. Veja, por exemplo, o retorno após o comando “2*A”.

(%i3) 2*A;

$$(\%o3)\quad \begin{bmatrix} 2 & 4 & 6 \\ 8 & 10 & 12 \\ 2\,a & 2\,b & 2\,c \end{bmatrix}$$

Insira outra matriz de nome B dada por $\begin{bmatrix} -1 & -2 & -3 \\ -4 & -5 & -6 \\ 1 & 1 & 1 \end{bmatrix}$.

Para adição de uma matriz com outra basta escrever o código de adição entre seus nomes. Veja a seguir a saída depois do comando “A+B”.

(%i13) A+B;

$$(\%o13)\quad \begin{bmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ a+1 & b+1 & c+1 \end{bmatrix}$$

Cuidado com a multiplicação de matrizes!

Como as matrizes A e B são 3x3 é possível multiplicá-las. Mas CUIDADO! O comando para multiplicação de matrizes é “.” E não “*”.

(%i14) A.B;

$$(\%o14)\quad \begin{bmatrix} -6 & -9 & -12 \\ -18 & -27 & -36 \\ c-4\,b-a & c-5\,b-2\,a & c-6\,b-3\,a \end{bmatrix}$$

O uso de “*” faz com que o Maxima multiplique coordenada a coordenada!

(%i15) A*B;

$$(\%o15)\quad \begin{bmatrix} -1 & -4 & -9 \\ -16 & -25 & -36 \\ a & b & c \end{bmatrix}$$

Esse tipo de confusão pode ser desastroso em seus trabalhos escolares.

Método de Eliminação de Gauss

Insira uma matriz B no Maxima, use o caminho Álgebra – Introduzir matriz...” ou faça a digitação diretamente na janela do software como mostrado a seguir.

```
(%i1) B:matrix([0,2,3,4],[-1,-4,2,3],[0,5,-2,-1],[2,9,-4,0]);
```

(%o1) $\begin{bmatrix} 0 & 2 & 3 & 4 \\ -1 & -4 & 2 & 3 \\ 0 & 5 & -2 & -1 \\ 2 & 9 & -4 & 0 \end{bmatrix}$

Podemos calcular o determinante da matriz com o uso do caminho “Álgebra – Determinante” ou escrevendo diretamente o comando como mostrado a seguir:

```
(%i2) determinant(B);
(%o2) -109
```

O primeiro elemento da primeira linha é nulo, devemos trocar a primeira linha por outra cujo primeiro elemento não seja nulo. Por exemplo, vamos trocar a primeira linha pela quarta. Atribuímos outro nome para a nova matriz, sugerimos o nome B1.

```
(%i3) B1:rowswap(B,1,4);
```

(%o3) $\begin{bmatrix} 2 & 9 & -4 & 0 \\ -1 & -4 & 2 & 3 \\ 0 & 5 & -2 & -1 \\ 0 & 2 & 3 & 4 \end{bmatrix}$

Temos agora um elemento não nulo na primeira posição da primeira linha. Vamos multiplicar essa linha pelo inverso desse elemento. No comando a seguir atribuímos à linha 1 de B1 o produto de $\frac{1}{2}$ por tal linha.

```
(%i4) B1[1]:(1/2)*B1[1];
```

(%o4) $[1, \frac{9}{2}, -2, 0]$

O Maxima retorna a primeira linha da matriz após a atribuição. Para ver como está a matriz B1 basta escrever seu nome.

```
(%i5) B1;
```

$$(\%o5)\quad \begin{bmatrix} 1 & \frac{9}{2} & -2 & 0 \\ -1 & -4 & 2 & 3 \\ 0 & 5 & -2 & -1 \\ 0 & 2 & 3 & 4 \end{bmatrix}$$

Agora vamos usar o comando “rowop(M,i,j,k)”. Esse comando faz a substituição da i-ésima linha pela adição da i-ésima linha com a linha “j” multiplicada por (-k). Atribuímos o nome B2 a essa nova matriz.

Como desejamos obter um elemento nulo na posição (2,1) devemos usar para k o valor “– 1”.

```
(%i6) B2:rowop(B1,2,1,-1);
```

$$(\%o6)\quad \begin{bmatrix} 1 & \frac{9}{2} & -2 & 0 \\ 0 & \frac{1}{2} & 0 & 3 \\ 0 & 5 & -2 & -1 \\ 0 & 2 & 3 & 4 \end{bmatrix}$$

Não será necessário anular os elementos abaixo do elemento (2,1), eles já são nulos.

Multipliquemos a segunda linha da matriz B2 por 2 para obtermos um primeiro elemento não nulo igual a 1.

```
(%i7) B2[2]:2*B2[2];
(%o7) [0,1,0,6]
```

A nova matriz B2 é a seguinte:

```
(%i8) B2;
```

$$(\%o8)\quad \begin{bmatrix} 1 & \frac{9}{2} & -2 & 0 \\ 0 & 1 & 0 & 6 \\ 0 & 5 & -2 & -1 \\ 0 & 2 & 3 & 4 \end{bmatrix}$$

Para zerar os elementos abaixo do elemento (2,2) vamos usar o comando “rowop(M,i,j,k)” conforme figuras a seguir.

Para zerar o elemento (3,2) faz-se o seguinte:

```
(%i9) B3:rowop(B2,3,2,5);
```

$$(\%o9)\quad \begin{bmatrix} 1 & \frac{9}{2} & -2 & 0 \\ 0 & 1 & 0 & 6 \\ 0 & 0 & -2 & -31 \\ 0 & 2 & 3 & 4 \end{bmatrix}$$

Para zerar o elemento (4,2) faz-se o seguinte:

```
(%i10) B4:rowop(B3,4,2,2);
```

$$(\%o10)\quad \begin{bmatrix} 1 & \frac{9}{2} & -2 & 0 \\ 0 & 1 & 0 & 6 \\ 0 & 0 & -2 & -31 \\ 0 & 0 & 3 & -8 \end{bmatrix}$$

Multiplicamos a terceira linha de B4 por $-\frac{1}{2}$.

```
(%i11) B4[3]:(-1/2)*B4[3];
```

$$(\%o11)\quad [0,0,1,\frac{31}{2}]$$

A matriz se torna:

```
(%i12) B4;
```

$$(\%o12)\quad \begin{bmatrix} 1 & \frac{9}{2} & -2 & 0 \\ 0 & 1 & 0 & 6 \\ 0 & 0 & 1 & \frac{31}{2} \\ 0 & 0 & 3 & -8 \end{bmatrix}$$

Para zerar o elemento (3,4) faz-se o seguinte:

```
(%i13) B5:rowop(B4,4,3,3);
```

$$(\%o13)\quad \begin{bmatrix} 1 & \frac{9}{2} & -2 & 0 \\ 0 & 1 & 0 & 6 \\ 0 & 0 & 1 & \frac{31}{2} \\ 0 & 0 & 0 & -\frac{109}{2} \end{bmatrix}$$

Finalmente multiplicamos a quarta linha da matriz B5 por $-\dfrac{2}{109}$ para que o primeiro elemento seja igual a 1.

```
(%i14) B5[4]:(-2/109)*B5[4];
(%o14) [0,0,0,1]
```

Como resultado final do Método de Eliminação de Gauss, temos a seguinte matriz.

```
(%i15) B5;
```

$$(\%o15)\quad \begin{bmatrix} 1 & \frac{9}{2} & -2 & 0 \\ 0 & 1 & 0 & 6 \\ 0 & 0 & 1 & \frac{31}{2} \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

Esse resultado pode ser comparado com o calculado diretamente pelo Maxima com o uso do comando “echelon(M)”.

```
(%i29) echelon(B);
```

$$(\%o29)\quad \begin{bmatrix} 1 & \frac{9}{2} & -2 & 0 \\ 0 & 1 & 0 & 6 \\ 0 & 0 & 1 & \frac{31}{2} \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

Inversão de matrizes por escalonamento

Pode-se utilizar o escalonamento para calcular a inversa de matrizes cujo determinante seja não nulo.
Para facilitar o desenvolvimento do exemplo, tomemos a mesma matriz B utilizada anteriormente. O leitor pode "copiar e colar" os comandos para fazer uma nova atribuição à matriz B.

```
(%i1) B:matrix([0,2,3,4],[-1,-4,2,3],[0,5,-2,-1],[2,9,-4,0]);
```

Deve-se aumentar a matriz B colocando-se uma matriz identidade de ordem 4 à sua direita.

```
(%i2) M:addcol(B,ident(4));
```

(%o2) $$\begin{bmatrix} 0 & 2 & 3 & 4 & 1 & 0 & 0 & 0 \\ -1 & -4 & 2 & 3 & 0 & 1 & 0 & 0 \\ 0 & 5 & -2 & -1 & 0 & 0 & 1 & 0 \\ 2 & 9 & -4 & 0 & 0 & 0 & 0 & 1 \end{bmatrix}$$

Vamos utilizar a mesma sequência de comandos anteriores para tornar a submatriz B de M uma submatriz escalonada na forma escada. O usuário pode "copiar e colar" os comandos já utilizados na mesma célula. Lembre-se de que para mudar de linha sem fazer a compilação o usuário deve usar o "Enter".

```
--> M1:rowswap(M,1,4);
    M1[1]:(1/2)*M1[1];
    M2:rowop(M1,2,1,-1);
    M2[2]:2*M2[2];
    M3:rowop(M2,3,2,5);
    M4:rowop(M3,4,2,2);
    M4[3]:(-1/2)*M4[3];
    M5:rowop(M4,4,3,3);
    M5[4]:(-2/109)*M5[4];
    M5;
```

Ao final das trans formações a matriz M5 será como a mostrada a seguir.

```
(%i14) M5;
```

(%o14) $$\begin{bmatrix} 1 & \frac{9}{2} & -2 & 0 & 0 & 0 & 0 & \frac{1}{2} \\ 0 & 1 & 0 & 6 & 0 & 2 & 0 & 1 \\ 0 & 0 & 1 & \frac{31}{2} & 0 & 5 & -\frac{1}{2} & \frac{5}{2} \\ 0 & 0 & 0 & 1 & -\frac{2}{109} & \frac{38}{109} & -\frac{3}{109} & \frac{19}{109} \end{bmatrix}$$

Vamos usar o elemento (2,2) para zerar o elemento (1,2).

```
(%i15) M6:rowop(M5,1,2,9/2);
```

$$(\%o15)\quad \begin{bmatrix} 1 & 0 & -2 & -27 & 0 & -9 & 0 & -4 \\ 0 & 1 & 0 & 6 & 0 & 2 & 0 & 1 \\ 0 & 0 & 1 & \frac{31}{2} & 0 & 5 & -\frac{1}{2} & \frac{5}{2} \\ 0 & 0 & 0 & 1 & -\frac{2}{109} & \frac{38}{109} & -\frac{3}{109} & \frac{19}{109} \end{bmatrix}$$

Vamos usar o elemento (3,3) para zerar o elemento (1,3).

```
(%i16) M7:rowop(M6,1,3,-2);
```

$$(\%o16)\quad \begin{bmatrix} 1 & 0 & 0 & 4 & 0 & 1 & -1 & 1 \\ 0 & 1 & 0 & 6 & 0 & 2 & 0 & 1 \\ 0 & 0 & 1 & \frac{31}{2} & 0 & 5 & -\frac{1}{2} & \frac{5}{2} \\ 0 & 0 & 0 & 1 & -\frac{2}{109} & \frac{38}{109} & -\frac{3}{109} & \frac{19}{109} \end{bmatrix}$$

Vamos usar o elemento (4,4) para zerar o elemento (1,4).

```
(%i17) M8:rowop(M7,1,4,4);
```

$$(\%o17)\quad \begin{bmatrix} 1 & 0 & 0 & 0 & \frac{8}{109} & -\frac{43}{109} & -\frac{97}{109} & \frac{33}{109} \\ 0 & 1 & 0 & 6 & 0 & 2 & 0 & 1 \\ 0 & 0 & 1 & \frac{31}{2} & 0 & 5 & -\frac{1}{2} & \frac{5}{2} \\ 0 & 0 & 0 & 1 & -\frac{2}{109} & \frac{38}{109} & -\frac{3}{109} & \frac{19}{109} \end{bmatrix}$$

Vamos usar o elemento (4,4) para zerar o elemento (2,4).

```
(%i18) M9:rowop(M8,2,4,6);
```

$$(\%o18)\quad \begin{bmatrix} 1 & 0 & 0 & 0 & \frac{8}{109} & -\frac{43}{109} & -\frac{97}{109} & \frac{33}{109} \\ 0 & 1 & 0 & 0 & \frac{12}{109} & -\frac{10}{109} & \frac{18}{109} & -\frac{5}{109} \\ 0 & 0 & 1 & \frac{31}{2} & 0 & 5 & -\frac{1}{2} & \frac{5}{2} \\ 0 & 0 & 0 & 1 & -\frac{2}{109} & \frac{38}{109} & -\frac{3}{109} & \frac{19}{109} \end{bmatrix}$$

Vamos usar o elemento (4,4) para zerar o elemento (3,4).

```
(%i19) M10:rowop(M9,3,4,31/2);
```

$$(\%o19)\quad \begin{bmatrix} 1 & 0 & 0 & 0 & \frac{8}{109} & -\frac{43}{109} & -\frac{97}{109} & \frac{33}{109} \\ 0 & 1 & 0 & 0 & \frac{12}{109} & -\frac{10}{109} & \frac{18}{109} & -\frac{5}{109} \\ 0 & 0 & 1 & 0 & \frac{31}{109} & -\frac{44}{109} & -\frac{8}{109} & -\frac{22}{109} \\ 0 & 0 & 0 & 1 & -\frac{2}{109} & \frac{38}{109} & -\frac{3}{109} & \frac{19}{109} \end{bmatrix}$$

A submatriz B de M transformou-se numa submatriz identidade. A inversa procurada é composta das colunas 5, 6, 7 e 8 da matriz M10. Para eliminar as colunas 1,2,3 e 4 usamos o seguinte comando:

```
(%i22) S:submatrix(M10,1,2,3,4);
```

$$(\%o22)\quad \begin{bmatrix} \frac{8}{109} & -\frac{43}{109} & -\frac{97}{109} & \frac{33}{109} \\ \frac{12}{109} & -\frac{10}{109} & \frac{18}{109} & -\frac{5}{109} \\ \frac{31}{109} & -\frac{44}{109} & -\frac{8}{109} & -\frac{22}{109} \\ -\frac{2}{109} & \frac{38}{109} & -\frac{3}{109} & \frac{19}{109} \end{bmatrix}$$

A matriz S é a inversa de B. Pode-se verificar isso mediante cálculo dos produtos:

```
(%i25) S.B;B.S;
```

$$(\%o25)\quad \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

$$(\%o26)\quad \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

O Maxima possui um comando que calcula a inversa de B, sua aplicação está a seguir.

```
(%i28) invert(B);
```

$$(\%o28)\quad \begin{bmatrix} \frac{8}{109} & -\frac{43}{109} & -\frac{97}{109} & \frac{33}{109} \\ \frac{12}{109} & -\frac{10}{109} & \frac{18}{109} & -\frac{5}{109} \\ \frac{31}{109} & -\frac{44}{109} & -\frac{8}{109} & -\frac{22}{109} \\ -\frac{2}{109} & \frac{38}{109} & -\frac{3}{109} & \frac{19}{109} \end{bmatrix}$$

Geração aleatória de matrizes

Para gerar aleatoriamente um número inteiro situado entre 0 e N, deve-se usar o comando "random(N+1)".

```
(%i1) random(33);
(%o1) 17
```

Para gerar números decimais entre 0 e N, deve-se usar um ponto flutuante no comando.

```
 -->   random(4.3);
(%o19) 0.71582038614906
```

No exemplo anterior o número aleatório estará entre 0 e 4.3. A quantidade de casas decimais utilizadas será o padrão em uso no Maxima.
Para gerar uma lista (ou vetor) aleatório pode-se o comando "random" dentro do comando "makelist".

```
(%i2) makelist(random(11), k, 1, 10);
(%o2) [3,10,9,4,7,0,6,5,4,2]
```

No exemplo anterior os números estarão entre 0 e 10.
Para gerar lista com números inteiros entre -10 e 10 pode-se fazer como no exemplo seguinte.

```
(%i3) makelist(random(11)*(-1)^random(2), k, 1, 10);
(%o3) [3,-8,-5,-3,-7,-5,8,3,7,6]
```

Para gerar matrizes com entradas aleatórias entre -10 e 10 pode-se escrever o comando “random(11)*(-1)^random(2)” em cada local da janela “Álgebra – Gerar matriz da expressão”.



A saída desse comando será:

```
(%i1) M: genmatrix(lambda([i,j], random(11)*(-1)^random(2)), 4, 4);
```

(%o1) $$\begin{bmatrix} 6 & -10 & -4 & 0 \\ -5 & -2 & 3 & 0 \\ 2 & 2 & -6 & -7 \\ 6 & -1 & -1 & -6 \end{bmatrix}$$

Com essas combinações o usuário poderá gerar matrizes aleatórias para praticar a técnica de Eliminação de Gauss.